////////////////////////////////////////////////////////////////////////////////
//
//  The MIT License
//
//  Copyright (c) 2008 - 2009. Dimarik
//
//  Permission is hereby granted, free of charge, to any person obtaining a copy
//  of this software and associated documentation files (the "Software"), to deal
//  in the Software without restriction, including without limitation the rights
//  to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
//  copies of the Software, and to permit persons to whom the Software is
//  furnished to do so, subject to the following conditions:
//
//  The above copyright notice and this permission notice shall be included in
//  all copies or substantial portions of the Software.
//
//  THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
//  IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
//  FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
//  AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
//  LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
//  OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN
//  THE SOFTWARE.
//
////////////////////////////////////////////////////////////////////////////////

package ru.goodcore.utils
{

import flash.external.ExternalInterface;
import flash.system.Capabilities;

/**
 * @return					NetworkingStatus code 
 *
 * @author					Dimarik
 * @version					1.0
 * @playerversion			Flash 9
 * @langversion				3.0
 * 
 * @see						ru.goodcore.utils.NetworkingSatus
 * @keyword					getnetworkingstatus, externalinterface
 */
	public function getNetworkingStatus():NetworkingStatus {

		if (_reason)
			return _reason;

		var playerType:	String = Capabilities.playerType;

		if (
			playerType == 'StandAlone' ||
			playerType == 'Desktop'
		) {
			_reason = NetworkingStatus.NONE;
		}

		else {
			try {
				ExternalInterface.call(null);
				_reason = NetworkingStatus.ALL;
			}
			catch (error:SecurityError) {
				_reason = NetworkingStatus.INTERNAL;
			}
			catch (error:Error) {
				// Error #2067: The ExternalInterface is not available in this container.
				// ExternalInterface requires Internet Explorer ActiveX, Firefox, 
				// Mozilla 1.7.5 and greater, or other browsers that support NPRuntime.
				_reason = NetworkingStatus.NONE;
			}
		}

		return _reason;
}
}

//=============================================================================
//
//  Inner definition
//
//=============================================================================

import ru.goodcore.utils.NetworkingStatus;

/**
 * @private
 * Possible reason values
 * 
 * null - Networking not yet checked
 * NetworkingSatus.ALL - Allow networking
 * NetworkingSatus.INTERNAL - Deny networking by security reason.
 * NetworkingSatus.NONE - No external interface found. Networking not possible.
 */
internal var _reason:  NetworkingStatus;